ELEMENTO CUADRILATERO - 4 NODOS

Implementación en Mathematica

v.2018

1. DATOS INICIALES

INICIO

In[1]:=
```
Off[General::spell1]
Off[General::spell]
```

In[9]:=
```
SetDirectory[NotebookDirectory[]]
```

Out[9]=
```
C:\#0-Modulos-M30x_MeF-10\#M308-m6-a5a-swm\11-I-cuadrilatero-i
```

DEFINICION NODOS - COORDENADAS NATURALES

DEFINICION GRAFICA

In[4]:=
```
CuaR4 = 
```

```
;
```

In[5]:=
```
CuaR4r = Show[CuaR4, ImageSize → 250]
```

Out[5]=



COORDENADAS NATURALES NODOS

In[13]:=
```
Cn = {{-1, -1}, {1, -1}, {1, 1}, {-1, 1}};
```

In[14]:=
```
NNodos = Dimensions[Cn][[1]]
```
Out[14]=
```
4
```

IMAGEN DEL ELEMENTO - COMPROBACION

FUNCION REPRESENTACION GRAFICA ELEMENTOS Y NODOS

In[15]:=
```
ElementPlot[b_List, options___] := Module[{asa, color, nr, circles, lines},
    asa = Select[{options}, ((! SameQ[#[[1]], NodeColor]) && (! SameQ[#[[1]], NodeSize])) &];
    {color, nr} = {NodeColor, NodeSize} /. {options} /.
      {NodeColor → GrayLevel[0], NodeSize → PointSize[0.06]};
    circles = Map[Point[#] &, Partition[Flatten[b], 2]];
    lines = Line[Append[b[[1]], First[b[[1]]]]];
    Show[Graphics[{nr, color, circles}], Graphics[lines], Evaluate[Sequence[##] &@@ asa]]];
```

DEFINICION VECTOR DE NODOS

In[16]:=
```
pts = {Cn[[1]], Cn[[2]], Cn[[3]], Cn[[4]]};
```

IMAGEN DE COMPROBACION

In[17]:=
```
Imagen = ElementPlot[{pts}, AspectRatio → 1, PlotRange → {{-1.5, 1.5}, {-1.5, 1.5}},
   ImageSize → 150, Frame → False, NodeColor → RGBColor[0, 0, 1]]
```
Out[17]=



3. FUNCIONES DE FORMA - METODO PRODUCTO DE CURVAS

CURVAS A CONSIDERAR

In[18]:=
```
Cu = Table[0, {i, 4}];
```

LADOS

In[19]:=
```
Cu[[1]] = (η + 1); Cu[[2]] = (ξ - 1); Cu[[3]] = (η - 1); Cu[[4]] = (ξ + 1);
```

DEFINICION PRODUCTOS DE CURVAS EN CADA NODO

In[20]:=
```
Nc = Table[0, {i, NNodos}];
```

Tipo 1 - ESQUINA

In[21]:=
```
Nc[[4]] = Cu[[1]] * Cu[[2]]
```
Out[21]=
$(1+\eta)\ (-1+\xi)$

In[22]:=
```
Nc[[3]] = Cu[[1]] * Cu[[4]]
```
Out[22]=
$(1+\eta)\ (1+\xi)$

In[23]:=
```
Nc[[2]] = Cu[[3]] * Cu[[4]]
```
Out[23]=
$(-1+\eta)\ (1+\xi)$

In[24]:=
```
Nc[[1]] = Cu[[2]] * Cu[[3]]
```
Out[24]=
$(-1+\eta)\ (-1+\xi)$

OBTENCION FUNCIONES DE FORMA

In[25]:=
```
Clear[Nf]
```
In[26]:=
```
Nfp = Table[0, {i, NNodos}];
```
In[27]:=
```
Nf = Table[0, {i, NNodos}];
```
In[28]:=
```
Do[
  Nfp[[i]] = a * Nc[[i]];
  eq = 1 == Nfp[[i]] /. {ξ -> Cn[[i, 1]], η -> Cn[[i, 2]]};
  as = a /. Solve[eq, a][[1]]; Print["Nodo ", i];
  Nf[[i]] = Simplify[Nfp[[i]] /. {a → as}],
  {i, NNodos}
 ];
```
Nodo 1

Nodo 2

Nodo 3

Nodo 4

In[29]:=
```
MatrixForm[Nf]
```
Out[29]//MatrixForm=

$$\begin{pmatrix} \frac{1}{4}\ (-1+\eta)\ (-1+\xi) \\ -\frac{1}{4}\ (-1+\eta)\ (1+\xi) \\ \frac{1}{4}\ (1+\eta)\ (1+\xi) \\ -\frac{1}{4}\ (1+\eta)\ (-1+\xi) \end{pmatrix}$$

COMPROBACION SUMA UNIDAD

In[30]:=

$$\text{Suma} = \sum_{i=1}^{\text{NNodos}} \text{Nf}[[i]]$$

Out[30]=

$$\frac{1}{4}(-1+\eta)(-1+\xi) - \frac{1}{4}(1+\eta)(-1+\xi) - \frac{1}{4}(-1+\eta)(1+\xi) + \frac{1}{4}(1+\eta)(1+\xi)$$

In[31]:=

```
Simplify[%]
```

Out[31]=

```
1
```

OK

REPRESENTACION GRAFICA.

Función Representación Gráfica Funciones de Forma

In[32]:=

```
PlotQuadrilateralShapeFunction[xyquad_, f_, Nsub_, aspect_] :=
  Module[{Ne, Nev, line3D = {}, poly3D = {}, xyf1, xyf2, xyf3, i, j, n,
    ixi, ieta, xi, eta, x1, x2, x3, x4, y1, y2, y3, y4, z1, z2, z3, z4, xc, yc},
   {{x1, y1, z1}, {x2, y2, z2}, {x3, y3, z3}, {x4, y4, z4}} = Take[xyquad, 4];
   xc = {x1, x2, x3, x4}; yc = {y1, y2, y3, y4};
   Ne[xi_, eta_] := N[{(1 - xi) * (1 - eta), (1 + xi) * (1 - eta), (1 + xi) * (1 + eta), (1 - xi) * (1 + eta)} / 4];
   n = Nsub; Do[Do[ixi = (2 * i - n - 1) / n; ieta = (2 * j - n - 1) / n;
     {xi, eta} = N[{ixi - 1 / n, ieta - 1 / n}]; Nev = Ne[xi, eta];
     xyf1 = {xc.Nev, yc.Nev, f[xi, eta]}; {xi, eta} = N[{ixi + 1 / n, ieta - 1 / n}]; Nev = Ne[xi, eta];
     xyf2 = {xc.Nev, yc.Nev, f[xi, eta]}; {xi, eta} = N[{ixi + 1 / n, ieta + 1 / n}]; Nev = Ne[xi, eta];
     xyf3 = {xc.Nev, yc.Nev, f[xi, eta]}; {xi, eta} = N[{ixi - 1 / n, ieta + 1 / n}]; Nev = Ne[xi, eta];
     xyf4 = {xc.Nev, yc.Nev, f[xi, eta]}; AppendTo[poly3D, Polygon[{xyf1, xyf2, xyf3, xyf4}]];
     AppendTo[line3D, Line[{xyf1, xyf2, xyf3, xyf4, xyf1}]], {i, 1, Nsub}], {j, 1, Nsub}];
   Show[Graphics3D[RGBColor[1, 1, 0]], Graphics3D[poly3D], Graphics3D[Thickness[.002]],
    Graphics3D[line3D], Graphics3D[RGBColor[0, 0, 0]], Graphics3D[Thickness[.005]],
    Graphics3D[Line[xyquad]], PlotRange → All, BoxRatios → {1, 1, aspect}, Boxed → False]];
```

Representación Gráfica Funciones Forma Elemento.

In[33]:=

```
Ng = Table[0, {i, NNodos}];
```

In[34]:=

```
xyc1 = {0, 0, 0}; xyc2 = {3, 0, 0}; xyc3 = {3, 3, 0};
xyc4 = {0, 3, 0}; xyquad = N[{xyc1, xyc2, xyc3, xyc4, xyc1}];
```

Control de Cuadricula

In[35]:=

```
Nsub = 10;
```

In[36]:=

```
Do[
  fi[ξ_, η_] = Nf[[i]];
  Ng[[i]] = PlotQuadrilateralShapeFunction[xyquad, fi, Nsub, 1 / 2],
  {i, NNodos}
 ];
```

4. RESULTADOS INTERACTIVOS

In[37]:=

```
Manipulate[{Imagen, Ng[[n]], Nf[[n]]}, {n, 1, Dimensions[Nf][[1]], 1}, {n, Range[Dimensions[Nf][[1]]]},
  FrameLabel → {"FUNCION DE FORMA EN NODO n - CUADRILATERO REGULAR 4 NODOS"}, SaveDefinitions → True]
```

Out[37]=

n

n 1 2 3 4

{  ,  , $\frac{1}{4}(-1+\eta)(-1+\xi)$ }

FUNCION DE FORMA EN NODO n – CUADRILATERO REGULAR 4 NODOS

5. DERIVADAS FUNCIONES DE FORMA Y JACOBIANO

FUNCIONES DE FORMA

In[38]:=

```
Nf
```

Out[38]=

$\left\{\frac{1}{4}(-1+\eta)(-1+\xi), -\frac{1}{4}(-1+\eta)(1+\xi), \frac{1}{4}(1+\eta)(1+\xi), -\frac{1}{4}(1+\eta)(-1+\xi)\right\}$

DERIVADAS FUNCIONES DE FORMA / COORDENADAS NATURALES

INICIALIZACIONES

In[39]:=

```
Clear[DNfξ, DNfη]
```

In[40]:=

```
DNfξ = Table[0, {i, NNodos}];
```

In[41]:=

```
DNfη = Table[0, {i, NNodos}];
```

PROCESO DE CALCULO

In[42]:=
```
Do[
  Nfe = Expand[Nf[[i]]];
  DNfξ[[i]] = Simplify[D[Nf[[i]], ξ]];
  DNfη[[i]] = Simplify[D[Nf[[i]], η]];
  Clear[Nfe],
  {i, NNodos}
 ];
```

RESULTADOS

In[43]:=
```
DNfξ
```

Out[43]= $\left\{\frac{1}{4}(-1+\eta), \frac{1-\eta}{4}, \frac{1+\eta}{4}, \frac{1}{4}(-1-\eta)\right\}$

In[44]:=
```
DNfη
```

Out[44]= $\left\{\frac{1}{4}(-1+\xi), \frac{1}{4}(-1-\xi), \frac{1+\xi}{4}, \frac{1-\xi}{4}\right\}$

MODULO CALCULO FUNCIONES DE FORMA Y JACOBIANO

MODULO GENERICO A COMPLETAR

MODULO GENERICO A COMPLETAR: XX = No. Nodos

```
(*QuadXXIsoPShapeFunDer[ncoor_,qcoor_]:=
  Module[{Nf,dNx,dNy,dNξ,dNη,J11,J12,J21,J22,Jdet,ξ,η,x,y},{ξ,η}=qcoor;
   Nf=(*Nf*);

   dNξ=(*dNξ*);

   dNη=(*dNη*);

   x=Table[ncoor[[i,1]],{i,XX}];y=Table[ncoor[[i,2]],{i,XX}];
   J11=dNξ.x;J21=dNξ.y;J12=dNη.x;J22=dNη.y;
   Jdet=Simplify[J11*J22-J12*J21];
   dNx=(J22*dNξ-J21*dNη)/Jdet;dNx=Simplify[dNx];
   dNy=(-J12*dNξ+J11*dNη)/Jdet;dNy=Simplify[dNy];
   Return[{Nf,dNx,dNy,Jdet}]];*)
```

MODULO COMPLETADO

In[45]:=

```
Quad4IsoPShapeFunDer[ncoor_, qcoor_] :=
  Module[{Nf, dNx, dNy, dNξ, dNη, J11, J12, J21, J22, Jdet, ξ, η, x, y}, {ξ, η} = qcoor;
   Nf = {1/4 (-1 + η) (-1 + ξ), -1/4 (-1 + η) (1 + ξ), 1/4 (1 + η) (1 + ξ), -1/4 (1 + η) (-1 + ξ)};

   dNξ = {1/4 (-1 + η), (1 - η)/4, (1 + η)/4, 1/4 (-1 - η)};

   dNη = {1/4 (-1 + ξ), 1/4 (-1 - ξ), (1 + ξ)/4, (1 - ξ)/4};

   x = Table[ncoor[[i, 1]], {i, 4}]; y = Table[ncoor[[i, 2]], {i, 4}];
   J11 = dNξ.x; J21 = dNξ.y; J12 = dNη.x; J22 = dNη.y;
   Jdet = Simplify[J11 * J22 - J12 * J21];
   dNx = (J22 * dNξ - J21 * dNη) / Jdet; dNx = Simplify[dNx];
   dNy = (-J12 * dNξ + J11 * dNη) / Jdet; dNy = Simplify[dNy];
   Return[{Nf, dNx, dNy, Jdet}]];
```

USO Y DESARROLLO

In[46]:=

```
{Nf, dNx, dNy, Jdet} = Quad4IsoPShapeFunDer[{{x1, y1}, {x2, y2}, {x3, y3}, {x4, y4}}, {ξ, η}];
```

In[47]:=

```
Jdet
```

Out[47]=

$$\frac{1}{8}(x1\,y2 - x3\,y2 - x1\,y4 + x3\,y4 - x3\,y1\,\eta - x1\,y2\,\eta + x1\,y3\,\eta + x3\,y4\,\eta + x3\,y1\,\xi - x3\,y2\,\xi - x1\,y3\,\xi + x1\,y4\,\xi + x4\,(y1 - y3\,(1+\eta) - y1\,\xi + y2\,(\eta+\xi)) + x2\,(y1\,(-1+\eta) + y3\,(1+\xi) - y4\,(\eta+\xi)))$$

In[48]:=

```
dNy[[1]]
```

Out[48]=

$$(x2 - x2\,\eta + x3\,(\eta-\xi) + x4\,(-1+\xi)) / (-x1\,y2 + x3\,y2 + x1\,y4 - x3\,y4 + x3\,y1\,\eta + x1\,y2\,\eta - x1\,y3\,\eta - x3\,y4\,\eta - x3\,y1\,\xi + x3\,y2\,\xi + x1\,y3\,\xi - x1\,y4\,\xi + x4\,(y3\,(1+\eta) + y1\,(-1+\xi) - y2\,(\eta+\xi)) + x2\,(y1 - y1\,\eta - y3\,(1+\xi) + y4\,(\eta+\xi)))$$

In[49]:=

```
coorn = {x1 → 0, y1 → 0, x2 → 1, y2 → 0, x3 → 1, y3 → 1 / 2, x4 → 0, y4 → 1 / 2}
```

Out[49]=

$$\{x1 \to 0,\ y1 \to 0,\ x2 \to 1,\ y2 \to 0,\ x3 \to 1,\ y3 \to \frac{1}{2},\ x4 \to 0,\ y4 \to \frac{1}{2}\}$$

In[50]:=

```
dNx /. coorn
```

Out[50]=

$$\left\{\frac{\frac{1}{2}-\frac{\xi}{2}+\frac{1}{2}(-\eta+\xi)}{-\frac{1}{2}-\frac{\eta}{2}+\frac{1}{2}(-1-\xi)+\frac{\eta+\xi}{2}},\ \frac{\frac{1}{2}(-1-\xi)+\frac{\eta+\xi}{2}}{-\frac{1}{2}-\frac{\eta}{2}+\frac{1}{2}(-1-\xi)+\frac{\eta+\xi}{2}},\ \frac{\frac{1}{2}+\frac{\eta}{2}}{\frac{1}{2}+\frac{\eta}{2}+\frac{1}{2}(-\eta-\xi)+\frac{1+\xi}{2}},\ \frac{1+\eta}{2\left(-\frac{1}{2}-\frac{\eta}{2}+\frac{1}{2}(-1-\xi)+\frac{\eta+\xi}{2}\right)}\right\}$$

6. MATRIZ DE RIGIDEZ - INTEGRACION NUMERICA

MODULO DE CALCULO DE LA MATRIZ DE RIGIDEZ

MODULO GENERICO A COMPLETAR

MODULO GENERICO: XX = No. Nodos, YY = Grados Libertad

```
(*QuadXXIsoPMembraneStiffness[ncoor_,mprop_,fprop_,options_]:=Module[{i,j,k,p=7,numer=False,
     Emat,th=1,h,qcoor,c,w,Nf,dNx,dNy,Jdet,B,Ke=Table[0,{YY},{YY}]},Emat=mprop[[1]];
    If[Length[options]==2,{numer,p}=options,{numer}=options];
    If[Length[fprop]>0,th=fprop[[1]]];
    If[p<1||p>4,Print["p out of range"];Return[Null]];
    For[k=1,k≤p*p,k++,
     {qcoor,w}=QuadGaussRuleInfo[{p,numer},k];
     {Nf,dNx,dNy,Jdet}=QuadXXIsoPShapeFunDer[ncoor,qcoor];
     If[Length[th]==0,h=th,h=th.Nf];c=w*Jdet*h;
     B={Flatten[Table[{dNx[[i]],0},{i,XX}]],
       Flatten[Table[{0,dNy[[i]]},{i,XX}]],Flatten[Table[{dNy[[i]],dNx[[i]]},{i,XX}]]};
     Ke+=Simplify[c*Transpose[B].(Emat.B)];];
    Return[Simplify[Ke]]];*)
```

MODULO COMPLETADO

In[51]:=

```
Quad4IsoPMembraneStiffness[ncoor_, mprop_, fprop_, options_] :=
  Module[{i, j, k, p = 7, numer = False, Emat, th = 1, h, qcoor, c,
    w, Nf, dNx, dNy, Jdet, B, Ke = Table[0, {8}, {8}]}, Emat = mprop[[1]];
   If[Length[options] == 2, {numer, p} = options, {numer} = options];
   If[Length[fprop] > 0, th = fprop[[1]]];
   If[p < 1 || p > 12, Print["p out of range"]; Return[Null]];
   For[k = 1, k ≤ p * p, k++,
    {qcoor, w} = QuadGaussRuleInfo[{p, numer}, k];
    {Nf, dNx, dNy, Jdet} = Quad4IsoPShapeFunDer[ncoor, qcoor];
    If[Length[th] == 0, h = th, h = th.Nf]; c = w * Jdet * h;
    B = {Flatten[Table[{dNx[[i]], 0}, {i, 4}]],
      Flatten[Table[{0, dNy[[i]]}, {i, 4}]], Flatten[Table[{dNy[[i]], dNx[[i]]}, {i, 4}]]};
    Ke += Simplify[c * Transpose[B].(Emat.B)];];
   Return[Simplify[Ke]]];
```

USO Y DESARROLLO

In[52]:=

```
B = {Flatten[Table[{dNx[[i]], 0}, {i, 4}]],
   Flatten[Table[{0, dNy[[i]]}, {i, 4}]], Flatten[Table[{dNy[[i]], dNx[[i]]}, {i, 4}]]};
```

In[53]:=

```
B // MatrixForm
```

Out[53]//MatrixForm=

$$\begin{pmatrix} \frac{y4+y2\ (-1+\eta)-y4\ \xi+y3\ (-\eta+\xi)}{-x1\ y2+x3\ y2+x1\ y4-x3\ y4+x3\ y1\ \eta+x1\ y2\ \eta-x1\ y3\ \eta-x3\ y4\ \eta-x3\ y1\ \xi+x3\ y2\ \xi+x1\ y3\ \xi-x1\ y4\ \xi+x4\ (y3\ (1+\eta)+y1\ (-1+\xi)-y2\ (\eta+\xi))+x2\ (y1-y1\ \eta-y3\ (1+\xi)+y4\ (\eta+\xi))} \\ 0 \\ \frac{x2-x2\ \eta+x3\ (\eta-\xi)+x4\ (-1+\xi)}{-x1\ y2+x3\ y2+x1\ y4-x3\ y4+x3\ y1\ \eta+x1\ y2\ \eta-x1\ y3\ \eta-x3\ y4\ \eta-x3\ y1\ \xi+x3\ y2\ \xi+x1\ y3\ \xi-x1\ y4\ \xi+x4\ (y3\ (1+\eta)+y1\ (-1+\xi)-y2\ (\eta+\xi))+x2\ (y1-y1\ \eta-y3\ (1+\xi)+y4\ (\eta+\xi))} \end{pmatrix}$$

In[54]:=
```
FullSimplify[dNx[[1]] /. coorn]
```

Out[54]=

$$\frac{1}{2}\,(-1+\eta)$$

In[55]:=
```
FullSimplify[B /. coorn] // MatrixForm
```

Out[55]//MatrixForm=

$$\begin{pmatrix} \frac{1}{2}(-1+\eta) & 0 & \frac{1-\eta}{2} & 0 & \frac{1+\eta}{2} & 0 & \frac{1}{2}(-1-\eta) & 0 \\ 0 & -1+\xi & 0 & -1-\xi & 0 & 1+\xi & 0 & 1-\xi \\ -1+\xi & \frac{1}{2}(-1+\eta) & -1-\xi & \frac{1-\eta}{2} & 1+\xi & \frac{1+\eta}{2} & 1-\xi & \frac{1}{2}(-1-\eta) \end{pmatrix}$$

In[56]:=
```
Emat = Em / (1 - nu^2) * {{1, nu, 0}, {nu, 1, 0}, {0, 0, (1 - nu) / 2}};
```

In[57]:=
```
Integrando = Jdet * h * Transpose[B].(Emat.B);
```

In[58]:=
```
matern = {Em → 96, nu → 1 / 3, h → 1}
```

Out[58]=

$$\left\{Em\to 96,\ nu\to \frac{1}{3},\ h\to 1\right\}$$

In[59]:=
```
FullSimplify[Integrando /. coorn /. matern] // MatrixForm
```

Out[59]//MatrixForm=

$$\begin{pmatrix} \frac{9}{8}(7+3(-2+\eta)\eta+4(-2+\xi)\xi) & \frac{9}{2}(-1+\eta)(-1+\xi) & -\frac{9}{8}\left(-1+3(-2+\eta)\eta+4\xi^2\right) & -\frac{9}{2}\,(\ldots \\ \frac{9}{2}(-1+\eta)(-1+\xi) & \frac{9}{8}(13+(-2+\eta)\eta+12(-2+\xi)\xi) & -\frac{9}{2}(-1+\eta)\xi & -\frac{9}{8}\,(-11+(\ldots \\ -\frac{9}{8}\left(-1+3(-2+\eta)\eta+4\xi^2\right) & -\frac{9}{2}(-1+\eta)\xi & \frac{9}{8}(7+3(-2+\eta)\eta+4\xi(2+\xi)) & \frac{9}{2}\,(-1\ldots \\ -\frac{9}{2}(-1+\eta)\xi & -\frac{9}{8}\left(-11+(-2+\eta)\eta+12\xi^2\right) & \frac{9}{2}(-1+\eta)(1+\xi) & \frac{9}{8}\,(13+(-2+\ldots \\ \frac{9}{8}\left(-7+3\eta^2+4\xi^2\right) & \frac{9}{2}(-1+\eta\xi) & -\frac{9}{8}\left(1+3\eta^2+4\xi(2+\xi)\right) & -\frac{9}{2}\ldots \\ \frac{9}{2}(-1+\eta\xi) & \frac{9}{8}\left(-13+\eta^2+12\xi^2\right) & -\frac{9}{2}\eta(1+\xi) & -\frac{9}{8}\,(11+\eta^2\ldots \\ -\frac{9}{8}\left(1+3\eta^2+4(-2+\xi)\xi\right) & -\frac{9}{2}\eta(-1+\xi) & \frac{9}{8}\left(-7+3\eta^2+4\xi^2\right) & \frac{9}{2}\,(\ldots \\ -\frac{9}{2}\eta(-1+\xi) & -\frac{9}{8}\left(11+\eta^2+12(-2+\xi)\xi\right) & \frac{9}{2}(1+\eta\xi) & \frac{9}{8}\,(-13\ldots \end{pmatrix}$$

In[60]:=

$$\texttt{FullSimplify}\left[\texttt{Integrando /. coorn /. matern /. }\left\{\xi\to-\frac{1}{\sqrt{3}},\ \eta\text{ -> }-\frac{1}{\sqrt{3}}\right\}\right]\texttt{ // MatrixForm}$$

Out[60]//MatrixForm=

$$\begin{pmatrix} \frac{21}{4}\left(2+\sqrt{3}\right) & 3\left(2+\sqrt{3}\right) & -\frac{3}{4}\left(2+3\sqrt{3}\right) & -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{21}{4} & -3 & -\frac{3}{4}\left(5+4\sqrt{3}\right) \\ 3\left(2+\sqrt{3}\right) & \frac{39}{4}\left(2+\sqrt{3}\right) & -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{3}{4}\left(-10+\sqrt{3}\right) & -3 & -\frac{39}{4} & -\frac{3}{2}\left(1+\sqrt{3}\right) \\ -\frac{3}{4}\left(2+3\sqrt{3}\right) & -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{3}{4}\left(-14+\sqrt{3}\right) & -3 & -\frac{15}{4}+3\sqrt{3} & \frac{3}{2}\left(-1+\sqrt{3}\right) & -\frac{21}{4} \\ -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{3}{4}\left(-10+\sqrt{3}\right) & -3 & \frac{1}{4}\left(78-33\sqrt{3}\right) & \frac{3}{2}\left(-1+\sqrt{3}\right) & -\frac{69}{4}+9\sqrt{3} & 6 \\ -\frac{21}{4} & -3 & -\frac{15}{4}+3\sqrt{3} & \frac{3}{2}\left(-1+\sqrt{3}\right) & -\frac{21}{4}\left(-2+\sqrt{3}\right) & 6-3\sqrt{3} & \frac{3}{4}\left(-2+3\sqrt{3}\right) \\ -3 & -\frac{39}{4} & \frac{3}{2}\left(-1+\sqrt{3}\right) & -\frac{69}{4}+9\sqrt{3} & 6-3\sqrt{3} & -\frac{39}{4}\left(-2+\sqrt{3}\right) & \frac{3}{2}\left(-1+\sqrt{3}\right) \\ -\frac{3}{4}\left(5+4\sqrt{3}\right) & -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{21}{4} & 6 & \frac{3}{4}\left(-2+3\sqrt{3}\right) & \frac{3}{2}\left(-1+\sqrt{3}\right) & \frac{3}{4}\left(14+\sqrt{3}\right) \\ -\frac{3}{2}\left(1+\sqrt{3}\right) & -\frac{69}{4}-9\sqrt{3} & 6 & -\frac{39}{4} & \frac{3}{2}\left(-1+\sqrt{3}\right) & \frac{3}{4}\left(10+\sqrt{3}\right) & -3 \end{pmatrix}$$

MODULO REGLAS DE CUADRATURA DE GAUSS

OPCION 2: DEFINICION DE CARLOS FELIPPA

In[61]:=

```
QuadGaussRuleInfo[{rule_, numer_}, point_] := Module[{xi, eta, p1, p2, i1, i2, w1, w2, k, info = Null},
    If[Length[rule] == 2, {p1, p2} = rule, p1 = p2 = rule];
    If[Length[point] == 2, {i1, i2} = point, k = point; i2 = Floor[(k - 1) / p1] + 1; i1 = k - p1 * (i2 - 1)];
    {xi, w1} = LineGaussRuleInfo[{p1, numer}, i1];
    {eta, w2} = LineGaussRuleInfo[{p2, numer}, i2];
    info = {{xi, eta}, w1 * w2};
    If[numer, Return[N[info]], Return[Simplify[info]]];];
```

In[62]:=

```
LineGaussRuleInfo[{rule_, numer_}, point_] :=
  Module[{g2 = {-1, 1} / Sqrt[3], w3 = {5 / 9, 8 / 9, 5 / 9}, g3 = {-Sqrt[3 / 5], 0, Sqrt[3 / 5]},
    w4 = {(1 / 2) - Sqrt[5 / 6] / 6, (1 / 2) + Sqrt[5 / 6] / 6, (1 / 2) + Sqrt[5 / 6] / 6, (1 / 2) - Sqrt[5 / 6] / 6},
    g4 = {-Sqrt[(3 + 2 * Sqrt[6 / 5]) / 7], -Sqrt[(3 - 2 * Sqrt[6 / 5]) / 7],
      Sqrt[(3 - 2 * Sqrt[6 / 5]) / 7], Sqrt[(3 + 2 * Sqrt[6 / 5]) / 7]}, g5 = {-Sqrt[5 + 2 * Sqrt[10 / 7]],
       -Sqrt[5 - 2 * Sqrt[10 / 7]], 0, Sqrt[5 - 2 * Sqrt[10 / 7]], Sqrt[5 + 2 * Sqrt[10 / 7]]} / 3,
    w5 = {322 - 13 * Sqrt[70], 322 + 13 * Sqrt[70], 512, 322 + 13 * Sqrt[70], 322 - 13 * Sqrt[70]} / 900,
    i = point, p = rule, info = {Null, 0}},
   If[p == 1, info = {0, 2}];
   If[p == 2, info = {g2[[i]], 1}];
   If[p == 3, info = {g3[[i]], w3[[i]]}];
   If[p == 4, info = {g4[[i]], w4[[i]]}];
   If[p == 5, info = {g5[[i]], w5[[i]]}];
   If[numer, Return[N[info]], Return[Simplify[info]]];
  ];
```

In[64]:=

```
QuadGaussRuleInfo[{2, False}, 1]
```

Out[64]=

$$\left\{\left\{-\frac{1}{\sqrt{3}}, -\frac{1}{\sqrt{3}}\right\}, 1\right\}$$

OPCION 1: DEFINICION EN MATHEMATICA

In[65]:=

```
<< NumericalDifferentialEquationAnalysis`;
```

In[66]:=

```
? GaussianQuadratureWeights
```

GaussianQuadratureWeights[n, a, b, prec] gives a list of the pairs {abscissa, weight} to prec digits precision for the elementary n–point Gaussian quadrature formula for quadrature on the interval a to b. The argument prec is optional. >>

In[67]:=

```
QuadGaussRuleInfo[{rule_, numer_}, point_] := Module[{xi, eta, p1, p2, i1, i2, w1, w2, k, info = Null},
    If[Length[rule] == 2, {p1, p2} = rule, p1 = p2 = rule];
    If[Length[point] == 2, {i1, i2} = point, k = point; i2 = Floor[(k - 1) / p1] + 1; i1 = k - p1 * (i2 - 1)];
    {xi, w1} = GaussianQuadratureWeights[p1, -1, 1][[i1]];
    {eta, w2} = GaussianQuadratureWeights[p2, -1, 1][[i2]];
    info = {{xi, eta}, w1 * w2};
    If[numer, Return[N[info]], Return[Simplify[info]]];
  ];
```

In[68]:=
```
QuadGaussRuleInfo[{25, False}, 1]
```
Out[68]=
```
{{-0.995557, -0.995557}, 0.000129819}
```

7. TEST DEL RECTANGULO

■ DEFINICION DE LA GEOMETRIA

In[69]:=
```
RectanguloT = 
```


```
;
```

In[71]:=
```
RectanguloTr = Show[RectanguloT, ImageSize → 250]
```
Out[71]=



■ DEFINCION COORDENADAS NODOS ELEMENTO - ANSYS CLASSIC

□ IMAGEN DEL MODELO EN ANSYS

In[72]:=
```
RectanguloA = 
```

```
;
```

In[74]:=
```
RectanguloAr = Show[RectanguloA, ImageSize → 450]
```
Out[74]=


DATOS COORDENADAS NODALES EN ANSYS

In[76]:=
```
Nodes = {{0.`, 0.`}, {1.`, 0.`}, {0.5`, 0.`},
    {1.`, 0.5`}, {1.`, 0.25`}, {0.`, 0.5`}, {0.5`, 0.5`}, {0.`, 0.25`}};
```
In[77]:=
```
numnod = Dimensions[DNodos][[1]]
```
Out[77]= 4

ORDENACION DATOS DE ANSYS

In[79]:=
```
CuaR4r
```
Out[79]=


In[80]:=
```
RectanguloAr
```

Out[80]=



In[81]:=
```
ncoor = {Nodes[[1]], Nodes[[2]], Nodes[[4]], Nodes[[6]]}
```

Out[81]=
```
{{0., 0.}, {1., 0.}, {1., 0.5}, {0., 0.5}}
```

DEFINICION DEL MATERIAL

In[82]:=
```
ClearAll[Em, nu, a, b, e, h, p, num]; h = 1;
```

```
Em = 96; nu = 1 / 3;  (*isotropic material*)
```

In[84]:=
```
Emat = Em / (1 - nu^2) * {{1, nu, 0}, {nu, 1, 0}, {0, 0, (1 - nu) / 2}};
```

In[85]:=
```
Emat // MatrixForm
```

Out[85]//MatrixForm=

$$\begin{pmatrix} 108 & 36 & 0 \\ 36 & 108 & 0 \\ 0 & 0 & 36 \end{pmatrix}$$

VERIFICACION DE LA MATRIZ DE RIGIDEZ

NUMERO DE PUNTOS DE GAUSS MINIMO PARA CONSEGUIR SUFICIENCIA DE RANGO

In[86]:=
```
NF = NNodos * 2.;
```

In[87]:=

$$NG = \frac{NF - 3}{3}$$

Out[87]=

1.66667

Se necesitan como mínimo 2 Puntos -- Regla 2 x 2 minima

BUCLE GENERICO A COMPLETAR

BUCLE GENERICO: XX = No. Nodos, ZZ = GRADOS DE LIBERTAD - 3

```
(*For [p=1, p≤5, p++,
 Ke=QuadXXIsoPMembraneStiffness[ncoor,{Emat,0,0},{h},{True,p}];
 Print["Gauss integration rule: ",p," x ",p];
 Print["Ke=",Chop[Ke]//MatrixForm];
 Valores=Chop[Eigenvalues[N[Ke]]];
 If[Valores[[ZZ]]≠0,Break[],Print["Valores proprios matriz Ke=",Valores]]
];
Print["Valores proprios matriz Ke=",Valores];
Print["tenemos la suficiencia de rango para p=",p]*)
```

DESARROLLO DE LA MATRIZ DE RIGIDEZ

In[88]:=

```
For [p = 1, p ≤ 5, p++,
  Ke = Quad4IsoPMembraneStiffness[ncoor, {Emat, 0, 0}, {h}, {True, p}];
  Print["Gauss integration rule: ", p, " x ", p];
  Print["Ke=", Chop[Ke] // MatrixForm];
  Valores = Chop[Eigenvalues[N[Ke]]];
  If[Valores[[5]] ≠ 0, Break[], Print["Valores proprios matriz Ke=", Valores]]
 ];
Print["Valores proprios matriz Ke=", Valores];
Print["tenemos la suficiencia de rango para p=", p]
```

Gauss integration rule: 1 x 1

$$Ke=\begin{pmatrix} 31.5 & 18. & 4.5 & 0 & -31.5 & -18. & -4.5 & 0 \\ 18. & 58.5 & 0 & 49.5 & -18. & -58.5 & 0 & -49.5 \\ 4.5 & 0 & 31.5 & -18. & -4.5 & 0 & -31.5 & 18. \\ 0 & 49.5 & -18. & 58.5 & 0 & -49.5 & 18. & -58.5 \\ -31.5 & -18. & -4.5 & 0 & 31.5 & 18. & 4.5 & 0 \\ -18. & -58.5 & 0 & -49.5 & 18. & 58.5 & 0 & 49.5 \\ -4.5 & 0 & -31.5 & 18. & 4.5 & 0 & 31.5 & -18. \\ 0 & -49.5 & 18. & -58.5 & 0 & 49.5 & -18. & 58.5 \end{pmatrix}$$

Valores proprios matriz Ke={223.64, 90., 46.3603, 0, 0, 0, 0, 0}

Gauss integration rule: 2 x 2

$$Ke=\begin{pmatrix} 42. & 18. & -6. & 0 & -21. & -18. & -15. & 0 \\ 18. & 78. & 0 & 30. & -18. & -39. & 0 & -69. \\ -6. & 0 & 42. & -18. & -15. & 0 & -21. & 18. \\ 0 & 30. & -18. & 78. & 0 & -69. & 18. & -39. \\ -21. & -18. & -15. & 0 & 42. & 18. & -6. & 0 \\ -18. & -39. & 0 & -69. & 18. & 78. & 0 & 30. \\ -15. & 0 & -21. & 18. & -6. & 0 & 42. & -18. \\ 0 & -69. & 18. & -39. & 0 & 30. & -18. & 78. \end{pmatrix}$$

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=2

In[91]:=

```
For [p = 3, p ≤ 7, p++,
  Ke = Quad4IsoPMembraneStiffness[ncoor, {Emat, 0, 0}, {h}, {True, p}];
  Print["Gauss integration rule: ", p, " x ", p];
  Print["Ke=", Chop[Ke] // MatrixForm];
  Valores = Chop[Eigenvalues[N[Ke]]];
  Print["Valores proprios matriz Ke=", Valores];
  Print["tenemos la suficiencia de rango para p=", p]
 ];
```

Gauss integration rule: 3 x 3

Ke=

42.	18.	-6.	0	-21.	-18.	-15.	0
18.	78.	0	30.	-18.	-39.	0	-69.
-6.	0	42.	-18.	-15.	0	-21.	18.
0	30.	-18.	78.	0	-69.	18.	-39.
-21.	-18.	-15.	0	42.	18.	-6.	0
-18.	-39.	0	-69.	18.	78.	0	30.
-15.	0	-21.	18.	-6.	0	42.	-18.
0	-69.	18.	-39.	0	30.	-18.	78.

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=3

Gauss integration rule: 4 x 4

Ke=

42.	18.	-6.	0	-21.	-18.	-15.	0
18.	78.	0	30.	-18.	-39.	0	-69.
-6.	0	42.	-18.	-15.	0	-21.	18.
0	30.	-18.	78.	0	-69.	18.	-39.
-21.	-18.	-15.	0	42.	18.	-6.	0
-18.	-39.	0	-69.	18.	78.	0	30.
-15.	0	-21.	18.	-6.	0	42.	-18.
0	-69.	18.	-39.	0	30.	-18.	78.

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=4

Gauss integration rule: 5 x 5

Ke=

42.	18.	-6.	0	-21.	-18.	-15.	0
18.	78.	0	30.	-18.	-39.	0	-69.
-6.	0	42.	-18.	-15.	0	-21.	18.
0	30.	-18.	78.	0	-69.	18.	-39.
-21.	-18.	-15.	0	42.	18.	-6.	0
-18.	-39.	0	-69.	18.	78.	0	30.
-15.	0	-21.	18.	-6.	0	42.	-18.
0	-69.	18.	-39.	0	30.	-18.	78.

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=5

Gauss integration rule: 6 x 6

Ke=

42.	18.	-6.	0	-21.	-18.	-15.	0
18.	78.	0	30.	-18.	-39.	0	-69.
-6.	0	42.	-18.	-15.	0	-21.	18.
0	30.	-18.	78.	0	-69.	18.	-39.
-21.	-18.	-15.	0	42.	18.	-6.	0
-18.	-39.	0	-69.	18.	78.	0	30.
-15.	0	-21.	18.	-6.	0	42.	-18.
0	-69.	18.	-39.	0	30.	-18.	78.

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=6

Gauss integration rule: 7 x 7

Ke=

42.	18.	-6.	0	-21.	-18.	-15.	0
18.	78.	0	30.	-18.	-39.	0	-69.
-6.	0	42.	-18.	-15.	0	-21.	18.
0	30.	-18.	78.	0	-69.	18.	-39.
-21.	-18.	-15.	0	42.	18.	-6.	0
-18.	-39.	0	-69.	18.	78.	0	30.
-15.	0	-21.	18.	-6.	0	42.	-18.
0	-69.	18.	-39.	0	30.	-18.	78.

Valores proprios matriz Ke={223.64, 90., 78., 46.3603, 42., 0, 0, 0}

tenemos la suficiencia de rango para p=7

8. TEST DEL TRAPECIO

DEFINICION DE LA GEOMETRIA

In[92]:=

TrapecioT =  ;

In[93]:=

```
TrapecioTr = Show[TrapecioT, ImageSize → 250]
```

Out[93]=



DEFINCION COORDENADAS NODOS ELEMENTO - ANSYS CLASSIC

IMAGEN DEL MODELO EN ANSYS

In[94]:=

TrapecioA =  ;

In[95]:=

```
TrapecioAr = Show[TrapecioA, ImageSize → 450]
```

Out[95]=



DATOS COORDENADAS NODALES EN ANSYS

In[97]:=

```
Nodes = {{0.`, 0.`}, {1.`, 0.`}, {0.5`, 0.`},
   {0.5`, 0.5`}, {0.75`, 0.25`}, {0.`, 0.5`}, {0.25`, 0.5`}, {0.`, 0.25`}};
```

In[98]:=

```
numnod = Dimensions[DNodos][[1]]
```

Out[98]=

4

ORDENACION DATOS DE ANSYS

In[99]:=

```
CuaR4r
```

Out[99]=



In[100]:=

```
TrapecioAr
```

Out[100]=



In[101]:=

```
ncoor = {Nodes[[1]], Nodes[[2]], Nodes[[4]], Nodes[[6]]}
```

Out[101]=

```
{{0., 0.}, {1., 0.}, {0.5, 0.5}, {0., 0.5}}
```

DEFINICION DEL MATERIAL

In[102]:=

```
ClearAll[Em, nu, a, b, e, h, p, num]; h = 1;
```

In[103]:=

```
Em = 96; nu = 1 / 3; (*isotropic material*)
```

In[104]:=

```
Emat = Em / (1 - nu^2) * {{1, nu, 0}, {nu, 1, 0}, {0, 0, (1 - nu) / 2}};
```

In[105]:=

```
Emat // MatrixForm
```

Out[105]//MatrixForm=

$$\begin{pmatrix} 108 & 36 & 0 \\ 36 & 108 & 0 \\ 0 & 0 & 36 \end{pmatrix}$$

VERIFICACION DE LA MATRIZ DE RIGIDEZ

NUMERO DE PUNTOS DE GAUSS MINIMO PARA CONSEGUIR SUFICIENCIA DE RANGO

In[106]:=

```
NF = NNodos * 2.;
```

In[107]:=

$$NG = \frac{NF - 3}{3}$$

Out[107]=

1.66667

Se necesitan como mínimo 2 Puntos -- Regla 2 x 2 minima

BUCLE GENERICO A COMPLETAR

BUCLE GENERICO: XX = No. Nodos, ZZ = GRADOS DE LIBERTAD - 3

In[108]:=

```
(*For [p=1, p≤5, p++,
 Ke=QuadXXIsoPMembraneStiffness[ncoor,{Emat,0,0},{h},{True,p}];
 Print["Gauss integration rule: ",p," x ",p];
 Print["Ke=",Chop[Ke]//MatrixForm];
 Valores=Chop[Eigenvalues[N[Ke]]];
 If[Valores[[ZZ]]≠0,Break[],Print["Valores proprios matriz Ke=",Valores]]
];
Print["Valores proprios matriz Ke=",Valores];
Print["tenemos la suficiencia de rango para p=",p]*)
```

DESARROLLO DE LA MATRIZ DE RIGIDEZ

In[109]:=

```
For [p = 1, p ≤ 5, p++,
  Ke = Quad4IsoPMembraneStiffness[ncoor, {Emat, 0, 0}, {h}, {True, p}];
  Print["Gauss integration rule: ", p, " x ", p];
  Print["Ke=", Chop[Ke] // MatrixForm];
  Valores = Chop[Eigenvalues[N[Ke]]];
  If[Valores[[5]] ≠ 0, Break[], Print["Valores proprios matriz Ke=", Valores]]
 ];
Print["Valores proprios matriz Ke=", Valores];
Print["tenemos la suficiencia de rango para p=", p]
```

Gauss integration rule: 1 x 1

$$Ke=\begin{pmatrix} 42. & 24. & -6. & -6. & -42. & -24. & 6. & 6. \\ 24. & 78. & -6. & 30. & -24. & -78. & 6. & -30. \\ -6. & -6. & 24. & -12. & 6. & 6. & -24. & 12. \\ -6. & 30. & -12. & 24. & 6. & -30. & 12. & -24. \\ -42. & -24. & 6. & 6. & 42. & 24. & -6. & -6. \\ -24. & -78. & 6. & -30. & 24. & 78. & -6. & 30. \\ 6. & 6. & -24. & 12. & -6. & -6. & 24. & -12. \\ 6. & -30. & 12. & -24. & -6. & 30. & -12. & 24. \end{pmatrix}$$

Valores proprios matriz Ke={200.216, 84., 51.7841, 0, 0, 0, 0, 0}

Gauss integration rule: 2 x 2

$$Ke=\begin{pmatrix} 47.0769 & 24.9231 & -11.0769 & -6.92308 & -31.8462 & -22.1538 & -4.15385 & 4.15385 \\ 24.9231 & 85.8462 & -6.92308 & 22.1538 & -22.1538 & -62.3077 & 4.15385 & -45.6923 \\ -11.0769 & -6.92308 & 29.0769 & -11.0769 & -4.15385 & 4.15385 & -13.8462 & 13.8462 \\ -6.92308 & 22.1538 & -11.0769 & 31.8462 & 4.15385 & -45.6923 & 13.8462 & -8.30769 \\ -31.8462 & -22.1538 & -4.15385 & 4.15385 & 62.3077 & 27.6923 & -26.3077 & -9.69231 \\ -22.1538 & -62.3077 & 4.15385 & -45.6923 & 27.6923 & 109.385 & -9.69231 & -1.38462 \\ -4.15385 & 4.15385 & -13.8462 & 13.8462 & -26.3077 & -9.69231 & 44.3077 & -8.30769 \\ 4.15385 & -45.6923 & 13.8462 & -8.30769 & -9.69231 & -1.38462 & -8.30769 & 55.3846 \end{pmatrix}$$

Valores proprios matriz Ke={203.335, 93.5194, 72.7045, 60.3702, 35.3013, 0, 0, 0}

tenemos la suficiencia de rango para p=2

In[112]:=

```
For [p = 3, p ≤ 7, p++,
  Ke = Quad4IsoPMembraneStiffness[ncoor, {Emat, 0, 0}, {h}, {True, p}];
  Print["Gauss integration rule: ", p, " x ", p];
  Print["Ke=", Chop[Ke] // MatrixForm];
  Valores = Chop[Eigenvalues[N[Ke]]];
  Print["Valores proprios matriz Ke=", Valores];
  Print["tenemos la suficiencia de rango para p=", p]
 ];
```

Gauss integration rule: 3 x 3

Ke=

47.1746	24.9524	-11.1746	-6.95238	-31.6508	-22.0952	-4.34921	4.09524
24.9524	85.9048	-6.95238	22.0952	-22.0952	-62.1905	4.09524	-45.8095
-11.1746	-6.95238	29.1746	-11.0476	-4.34921	4.09524	-13.6508	13.9048
-6.95238	22.0952	-11.0476	31.9048	4.09524	-45.8095	13.9048	-8.19048
-31.6508	-22.0952	-4.34921	4.09524	62.6984	27.8095	-26.6984	-9.80952
-22.0952	-62.1905	4.09524	-45.8095	27.8095	109.619	-9.80952	-1.61905
-4.34921	4.09524	-13.6508	13.9048	-26.6984	-9.80952	44.6984	-8.19048
4.09524	-45.8095	13.9048	-8.19048	-9.80952	-1.61905	-8.19048	55.619

Valores proprios matriz Ke={203.402, 93.9307, 73.0147, 60.8077, 35.6384, 0, 0, 0}

tenemos la suficiencia de rango para p=3

Gauss integration rule: 4 x 4

Ke=

47.1776	24.9533	-11.1776	-6.95327	-31.6449	-22.0935	-4.35514	4.09346
24.9533	85.9065	-6.95327	22.0935	-22.0935	-62.1869	4.09346	-45.8131
-11.1776	-6.95327	29.1776	-11.0467	-4.35514	4.09346	-13.6449	13.9065
-6.95327	22.0935	-11.0467	31.9065	4.09346	-45.8131	13.9065	-8.18692
-31.6449	-22.0935	-4.35514	4.09346	62.7103	27.8131	-26.7103	-9.81308
-22.0935	-62.1869	4.09346	-45.8131	27.8131	109.626	-9.81308	-1.62617
-4.35514	4.09346	-13.6449	13.9065	-26.7103	-9.81308	44.7103	-8.18692
4.09346	-45.8131	13.9065	-8.18692	-9.81308	-1.62617	-8.18692	55.6262

Valores proprios matriz Ke={203.404, 93.9434, 73.0241, 60.821, 35.6484, 0, 0, 0}

tenemos la suficiencia de rango para p=4

Gauss integration rule: 5 x 5

Ke=

47.1777	24.9533	-11.1777	-6.9533	-31.6447	-22.0934	-4.35532	4.0934
24.9533	85.9066	-6.9533	22.0934	-22.0934	-62.1868	4.0934	-45.8132
-11.1777	-6.9533	29.1777	-11.0467	-4.35532	4.0934	-13.6447	13.9066
-6.9533	22.0934	-11.0467	31.9066	4.0934	-45.8132	13.9066	-8.18681
-31.6447	-22.0934	-4.35532	4.0934	62.7106	27.8132	-26.7106	-9.81319
-22.0934	-62.1868	4.0934	-45.8132	27.8132	109.626	-9.81319	-1.62638
-4.35532	4.0934	-13.6447	13.9066	-26.7106	-9.81319	44.7106	-8.18681
4.0934	-45.8132	13.9066	-8.18681	-9.81319	-1.62638	-8.18681	55.6264

Valores proprios matriz Ke={203.404, 93.9437, 73.0244, 60.8214, 35.6487, 0, 0, 0}

tenemos la suficiencia de rango para p=5

Gauss integration rule: 6 x 6

Ke=

47.1777	24.9533	-11.1777	-6.9533	-31.6447	-22.0934	-4.35532	4.0934
24.9533	85.9066	-6.9533	22.0934	-22.0934	-62.1868	4.0934	-45.8132
-11.1777	-6.9533	29.1777	-11.0467	-4.35532	4.0934	-13.6447	13.9066
-6.9533	22.0934	-11.0467	31.9066	4.0934	-45.8132	13.9066	-8.18681
-31.6447	-22.0934	-4.35532	4.0934	62.7106	27.8132	-26.7106	-9.81319
-22.0934	-62.1868	4.0934	-45.8132	27.8132	109.626	-9.81319	-1.62639
-4.35532	4.0934	-13.6447	13.9066	-26.7106	-9.81319	44.7106	-8.18681
4.0934	-45.8132	13.9066	-8.18681	-9.81319	-1.62639	-8.18681	55.6264

Valores proprios matriz Ke={203.404, 93.9438, 73.0244, 60.8214, 35.6487, 0, 0, 0}

tenemos la suficiencia de rango para p=6

Gauss integration rule: 7 x 7

$$Ke=\begin{pmatrix} 47.1777 & 24.9533 & -11.1777 & -6.9533 & -31.6447 & -22.0934 & -4.35532 & 4.0934 \\ 24.9533 & 85.9066 & -6.9533 & 22.0934 & -22.0934 & -62.1868 & 4.0934 & -45.8132 \\ -11.1777 & -6.9533 & 29.1777 & -11.0467 & -4.35532 & 4.0934 & -13.6447 & 13.9066 \\ -6.9533 & 22.0934 & -11.0467 & 31.9066 & 4.0934 & -45.8132 & 13.9066 & -8.18681 \\ -31.6447 & -22.0934 & -4.35532 & 4.0934 & 62.7106 & 27.8132 & -26.7106 & -9.81319 \\ -22.0934 & -62.1868 & 4.0934 & -45.8132 & 27.8132 & 109.626 & -9.81319 & -1.62639 \\ -4.35532 & 4.0934 & -13.6447 & 13.9066 & -26.7106 & -9.81319 & 44.7106 & -8.18681 \\ 4.0934 & -45.8132 & 13.9066 & -8.18681 & -9.81319 & -1.62639 & -8.18681 & 55.6264 \end{pmatrix}$$

Valores proprios matriz Ke={203.404, 93.9438, 73.0244, 60.8214, 35.6487, 0, 0, 0}

tenemos la suficiencia de rango para p=7